;     Функционално програмиране.

;     Упражнение N14

;Задача 1. Табулацията е техника, която позволява на функция да записва в
;локална таблица стойности, които вече са пресметнати. Да се напише
;функция, която пресмята стойностите на една функция f чрез табулация.

(require (lib "list.ss"))
(require (lib "compat.ss"))
(define (fib n)

  (cond ((= n 0) 0)

        ((= n 1) 1)

        (else (+ (fib (- n 1))

                 (fib (- n 2))))))

(define (lookup key table)

  (let ((record (assoc key (cdr table))))

    (if (not record)

        ()
        (cdr record))))

(define (insert! key value table)
  (let ((record (assoc key (cdr table))))
    (if (not record)

        (set-cdr! table

                  (cons (cons key value) (cdr table)))

        (set-cdr! record value)))

  'ok)

(define (make-table)

  (list '*table*))

(define (memoize f)

  (let ((table (make-table)))

    (lambda (x)

      (let ((previously-computed-result (lookup x table)))

        (if (not (null? previously-computed-result))

            previously-computed-result

            (let ((result (f x)))

              (insert! x result table)

              result))))))

(define memo-fib           ;(memo-fib 10) -> 55

  (memoize (lambda (n)

             (cond ((= n 0) 0)

                   ((= n 1) 1)

                   (else (+ (memo-fib (- n 2))

                            (memo-fib (- n 1))))))))

;Задача 2. Accumulator е функция, която се използва с един числов

;аргумент и прибавя аргумента си към сумата. Всеки път, когато тя

;се извиква, тя връща текущата натрупана сума. Да се напише

;функция make-accumulator, която генерира акумулатор. Входът към

;make-accumulator трябва да указва началната стойност на сумата,

;напр. (define A (make-accumulator 5)), (A 10) -> 15, (A 10) -> 25

(define (make-accumulator balance)

  (lambda (amount)

    (begin (set! balance (+ balance amount))

              balance)))

;Задача 3. Да се напише функция make-monitored (преброява колко

;пъти една функция е извиквана в процеса на изчисление) с вход

;функция f, която има един вход. Резултатът от make-monitored e

;трета функция с име mf, която пази броя на повикванията чрез

;вътрешен брояч. Ако вход на mf e how-many-calls?, тогава mf връща

;стойността на брояча. Ако входът е reset-count, тогава mf

;установява брояча на 0. За всеки друг вход mf връща резултата от

;извикването на f с този вход и увеличава брояча. Напр.

;(define s (make-monitored sqrt)),

;(s 100) -> 10, (s 'how-many-calls?) -> 1

(define (make-monitored f)

  (define count 0)

  (define (mf m)

    (cond ((eq? m 'how-many-calls?) count)

          ((eq? m 'reset-count) (set! count 0))

          (else (begin (set! count (+ count 1))  (f m)))))

  mf)

;Задача 4. Да разгледаме функцията make-cycle.

(define (make-cycle x)

  (set-cdr! (last-pair x) x)

  x)

;Вградената функция (last-pair x) връща списък от последния

;елемент в нейния аргумент, който трябва да е списък.

;Нека (define z (make-cycle '(a b c))). Какво ще се случи, ако

;изчислим (last-pair z) и защо? 

;Задача 5. Дадена е следната функция за преброяване на точковите

;двойки, от които е съставена всяка списъчна структура.

;(define (count-pairs x)

;  (if (atom? x)

;      0

;      (+ (count-pairs (car x))

;         (count-pairs (cdr x))

;         1)))

;Да се покаже, че не работи правилно. Да се намерят примери за

;структури от точно три двойки, за които функцията връща 3 ; 4 ;

;7, не връща нищо.

;(count-pairs '(a b c)) -> 3

;(define x '(a b c)), (set-car! x (last-pair x)), x -> ((c) b c),

;(count-pairs x) -> 4

;(define x '(a b c)), (set-car! (cdr x) (last-pair x)),

;x -> (a (c) c),

;(set-car! x (cdr x)), x -> (((c) c) (c) c), (count-pairs x) -> 7

;(count-pairs (make-cycle '(a b c))) -> не връща нищо.

;Задача 6. Да се напише правилна версия на count-pairs, която

;връща броя на различните двойки във всяка структура. Да се

;поддържа спомагателна структура, която пази следа кои двойки са

;преброени.

(define (count-pairs l)    ;В pairs се включват всички двойки

  (let ((pairs (list l)))  ;(клетки от представянето на l), които

    (define (count x)      ;вече са достигнати. l е първата клетка,

      (if (atom? x)        ;а към нея може да има указател(цикъл).

          0                ;В началото попадаме в l(указателя към

          (+ (cond ((memq (car x) pairs) 0) ;първата клетка от

                                      ;представянето на списъка).

                   (else (set! pairs (cons (car x) pairs))

                         (count (car x))))

             (cond ((memq (cdr x) pairs) 0)

                   (else (set! pairs (cons (cdr x) pairs))

                         (count (cdr x))))

             1)))

    (count l)))

;Задача 7. Да се напише функция, която проверява списък и определя

;дали съдържа цикъл.

(define (cycled? l)

  (let ((elements (list l)))

    (define (traverse x)

      (cond ((atom? x) #F)

            ((memq (car x) elements) #T)

            ((memq (cdr x) elements) #T)

            (else (set! elements (cons (car x) (cons (cdr x) elements)))

                  (or (traverse (car x))

                      (traverse (cdr x))))))

    (traverse l)))

;(define l '(a b c)), (set-car! l l), (cycled? l) -> #T

;(define l '(a (b) c)), (set-car! (cadr l) (cdr l)), (cycled? l) -> #T

;(define l '(a)), (set-car! l l), (cycled? l) -> #T

